

Math 252 — Fall 2002

Answers to Matrix Exponential Exercises

Answers, using Maple (you should do the problems by hand, but it is useful to know how to check your answers using Maple):

First introduce a package that understands matrices. The new `LinearAlgebra` package is used to exploit its more natural (if somewhat verbose) set of function names. Once the package is loaded, the matrices appearing in the exercises can be defined. Since the matrices are small, it is equally easy to enter them by rows or by columns. A mixture of both styles is used to illustrate the use of these **shortcuts** for defining matrices. Almost all matrices are called A in the problem statements, so we do the same here, modified by information about the source of the problem. Note that we ask Maple to find $A + B$ in problem 5.

```
with(LinearAlgebra):
A1a := < <1 | 1>, <0 | 0> >;
A1b := < <5, -1> | <6, -2> >;
A1c := < <2, 1> | <-8, -4> >;
A1d := < <2, 0, 0> | <2, 1, 0> | <1, 2, -1> >;
A4 := < <0, 0, 0> | <1, 0, 0> | <2, 1, 0> >;
c5 := <1, 0>; A5 := <c5 | c5>;
B5 := ZeroMatrix(2, compact=false); B5[1, 2] := -1;
AplusB5 := A5 + B5;
```

Here is the response

$$\begin{aligned}A1a &:= \begin{bmatrix} 1 & 1 \\ 0 & 0 \end{bmatrix} \\A1b &:= \begin{bmatrix} 5 & 6 \\ -1 & -2 \end{bmatrix} \\A1c &:= \begin{bmatrix} 2 & -8 \\ 1 & -4 \end{bmatrix} \\A1d &:= \begin{bmatrix} 2 & 2 & 1 \\ 0 & 1 & 2 \\ 0 & 0 & -1 \end{bmatrix} \\A4 &:= \begin{bmatrix} 0 & 1 & 2 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix} \\c5 &:= \begin{bmatrix} 1 \\ 0 \end{bmatrix} \\A5 &:= \begin{bmatrix} 1 & 1 \\ 0 & 0 \end{bmatrix} \\B5 &:= \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \\B5_{1,2} &:= -1 \\AplusB5 &:= \begin{bmatrix} 1 & 0 \\ 0 & 0 \end{bmatrix}\end{aligned}$$

Only the steps in the construction of $B5$ are shown, but you can check that the correct result has been found by entering $B5$; on the command line. For all the matrices in problem 1, the eigenvalues and eigenvectors are found. The new package produces the results in a directly usable form. The trick of assigning to a list is described in examples on the help page for this operation. When the matrices are built, we check the results by computing AS and $S\Lambda$ side by side.

```
(L1a,S1a):=Eigenvectors(A1a);Lambda1a:=DiagonalMatrix(L1a);
(L1b,S1b):=Eigenvectors(A1b);Lambda1b:=DiagonalMatrix(L1b);
(L1c,S1c):=Eigenvectors(A1c);Lambda1c:=DiagonalMatrix(L1c);
(L1d,S1d):=Eigenvectors(A1d);Lambda1d:=DiagonalMatrix(L1d);
A1a.S1a,S1a.Lambda1a;
A1b.S1b,S1b.Lambda1b;
A1c.S1c,S1c.Lambda1c;
A1d.S1d,S1d.Lambda1d;
```

Here is the response

$$\begin{aligned}
 L1a, S1a &:= \begin{bmatrix} 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 & 1 \\ 1 & 0 \end{bmatrix} \\
 Lambda1a &:= \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix} \\
 L1b, S1b &:= \begin{bmatrix} 4 \\ -1 \end{bmatrix}, \begin{bmatrix} -5 & 1 \\ 1 & -1 \end{bmatrix} \\
 Lambda1b &:= \begin{bmatrix} 4 & 0 \\ 0 & -1 \end{bmatrix} \\
 L1c, S1c &:= \begin{bmatrix} -2 \\ 0 \end{bmatrix}, \begin{bmatrix} 2 & 4 \\ 1 & 1 \end{bmatrix} \\
 Lambda1c &:= \begin{bmatrix} -2 & 0 \\ 0 & 0 \end{bmatrix} \\
 L1d, S1d &:= \begin{bmatrix} 1 \\ -1 \\ 2 \end{bmatrix}, \begin{bmatrix} -2 & 1 & 1 \\ 1 & -3 & 0 \\ 0 & 3 & 0 \end{bmatrix} \\
 Lambda1d &:= \begin{bmatrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 2 \end{bmatrix} \\
 \begin{bmatrix} 0 & 1 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 \\ 0 & 0 \end{bmatrix} \\
 \begin{bmatrix} -24 & -1 \\ 4 & 1 \end{bmatrix}, \begin{bmatrix} -24 & -1 \\ 4 & 1 \end{bmatrix} \\
 \begin{bmatrix} -4 & 0 \\ -2 & 0 \end{bmatrix}, \begin{bmatrix} -4 & 0 \\ -2 & 0 \end{bmatrix}
 \end{aligned}$$

$$\begin{bmatrix} -2 & -1 & 2 \\ 1 & 3 & 0 \\ 0 & -3 & 0 \end{bmatrix}, \begin{bmatrix} -2 & -1 & 2 \\ 1 & 3 & 0 \\ 0 & -3 & 0 \end{bmatrix}$$

This completes the calculation of the quantities requested in the first problem. One should check that the factorization represents the given quantity. Then, a modification of Λ in this factorization will find the quantities requested in problems 2 and 3. One must rebuild a temporary copy of Λ in these examples because this `Map` command does its work in place. We show the instructions for part (a) only, but give all answers.

```
Ans1a:=Equal(A1a,S1a.Lambda1a.S1a^(-1) #Problem 1
X:=DiagonalMatrix(L1a): Map(x->x^6,X): Ans2a:=S1a.X.S1a^(-1); #Problem 2
X:=DiagonalMatrix(L1a): Map(x->exp(t*x),X): Ans3a:=S1a.X.S1a^(-1); #Problem 3
```

Here is a typical response in problem 1 and all the responses to problems 2 and 3.

$$\begin{aligned} \text{Ans1a} &:= \text{true} \\ \text{Ans2a} &:= \begin{bmatrix} 1 & 1 \\ 0 & 0 \end{bmatrix} \\ \text{Ans2b} &:= \begin{bmatrix} 4915 & 4914 \\ -819 & -818 \end{bmatrix} \\ \text{Ans2c} &:= \begin{bmatrix} -64 & 256 \\ -32 & 128 \end{bmatrix} \\ \text{Ans2d} &:= \begin{bmatrix} 64 & 126 & 105 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \\ \text{Ans3a} &:= \begin{bmatrix} e^t & -1 + e^t \\ 0 & 1 \end{bmatrix} \\ \text{Ans3b} &:= \begin{bmatrix} \frac{6}{5}e^{4t} - \frac{1}{5}e^{-t} & \frac{6}{5}e^{4t} - \frac{6}{5}e^{-t} \\ -\frac{1}{5}e^{4t} + \frac{1}{5}e^{-t} & -\frac{1}{5}e^{4t} + \frac{6}{5}e^{-t} \end{bmatrix} \\ \text{Ans3c} &:= \begin{bmatrix} -e^{2t} + 2 & 4e^{2t} - 4 \\ -\frac{1}{2}e^{2t} + \frac{1}{2} & 2e^{2t} - 1 \end{bmatrix} \\ \text{Ans3d} &:= \begin{bmatrix} e^{2t} & -2e^t + 2e^{2t} & -2e^t + \frac{1}{3}e^{-t} + \frac{5}{3}e^{2t} \\ 0 & e^t & e^t + e^{-t} \\ 0 & 0 & e^{-t} \end{bmatrix} \end{aligned}$$

These matrices can be checked: for example, `subs(t=0,Ans3a)` evaluates the exponential `Ans3a` at $t = 0$, so should give an identity matrix. The derivative can be found by `map(diff,Ans3a,t)` — the lower case `map` will give a new matrix and not overwrite `Ans3a` — and this should be the same as `A1a.Ans3a`.

Moving on to Problem 4, $A4$ is a triangular matrix with diagonal zero, so some power will be a zero matrix and e^{A4} will be a polynomial. The following instructions verify that $A^3 = 0$ and then sums the nonzero terms of the Taylor series.

```

A4p2:=A4.A4;
A4p3:=A4.A4p2;
ExpA4:=1+map(x->t*x,A4)+map(x->t^2*x/2,A4p2);

```

Here are the results.

$$\begin{aligned}
 A4p2 &:= \begin{bmatrix} 0 & 0 & 1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \\
 A4p3 &:= \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \\
 ExpA4 &:= \begin{bmatrix} 1 & t & 2t + \frac{1}{2}t^2 \\ 0 & 1 & t \\ 0 & 0 & 1 \end{bmatrix}
 \end{aligned}$$

We have seen all the methods for finding e^{At} , so we can give the results without showing the Maple instructions.

$$\begin{aligned}
 Ans5A &:= \begin{bmatrix} e^t & -1 + e^t \\ 0 & 1 \end{bmatrix} \\
 Ans5B &:= \begin{bmatrix} 1 & -t \\ 0 & 1 \end{bmatrix} \\
 Ans5AplusB &:= \begin{bmatrix} e^t & 0 \\ 0 & 1 \end{bmatrix} \\
 Ans5A.Ans5B &:= \begin{bmatrix} e^t & -1 + e^t - te^t \\ 0 & 1 \end{bmatrix}
 \end{aligned}$$

The results for $e^{At}e^{Bt}$ and $e^{(A+B)t}$ are **visibly** different.